

## Работа с файлами в PascalABC.NET.

Файл представляет собой последовательность элементов одного типа, хранящихся на диске. В **PascalABC.NET** имеется два типа файлов - *двоичные* и *текстовые*. Текстовые файлы хранят символы, разделенные на строки символами #13#10 (Windows) и символом #10 (Linux). Двоичные файлы в свою очередь делятся на типизированные и бестиповые.

Для описания текстового файла используется стандартное имя типа `text`, бестиповые файлы имеют тип `file`, а для описания типизированного файла используется конструкция `file of` тип элементов:

```
var
  f1: file of real;
  f2: text;
  f3: file;
```

Некоторые процедуры и функции для работы с файлами через файловые переменные:

`Assign(f: FileType, name: string);` Связывает файловую переменную `f` с именем файла `name`.

`Close(f: FileType);` Закрывает файл `f`.

`Reset(f: Text);` Открывает текстовый файл `f` на чтение.

`Rewrite(f: Text);` Открывает текстовый файл `f` на запись, обнуляя его содержимое. Если файл существовал, он обнуляется.

`Append(f: Text);` Открывает текстовый `f` файл на дополнение.

Функция `Eof(f: FileType): boolean;` Возвращает `True`, если достигнут конец файла.

Пример программы с использованием файлов. Мы просим указать количество дробных чисел, которые определяются случайным образом и записываются в файл `output.txt`. Его предварительно создадим в Блокнотом в папке, где мы сохраним программу. Вторая часть программы читает файл неизвестной длины построчно и выводит каждую считанную из файла строку на экран. Преобразовав строку содержащую число, в число пригодное для математических операций.

```
var f:Text; // объявляем файловую переменную
    a:string;
begin
  Assign(f, 'output.txt'); //связываем эту переменную с файлом на диске
  Rewrite(f); // открываем файл для записи
  var n := ReadInteger('Введите n:');
  for var i:=1 to n do writeln(f, (random(201)-100)/10);
  close (f); // закрываем файл
  // файл создан
  Reset(f); // открываем этот же файл для чтения
  while not eof(f) do // пока не достигнут конец файла...
    begin
      readln(f,a); // читаем очередную строку (содержащую число)
      writeln(StrToFloat(a):10:5); // преобразуем строку в число и
      // выводим на экран
    end;
  close(f); // закрываем файл
end.
```

Результат работы программы:

```
Окно вывода
Введите n: 9
-8.80000
0.30000
1.80000
2.40000
4.90000
-2.20000
-6.00000
-4.10000
-1.00000
```

А так выглядит файл output.txt, открытый блокнотом:

```
output - Блокнот
Файл  Правка  Формат  Вид  Справка
|-8.8
0.3
1.8
2.4
4.9
-2.2
-6
-4.1
-1
```

Если мы точно знаем количество чисел (строк) в файле, то это число имеет смысл написать в первой строке создаваемого файла с данными.

```
var f:Text;
    a:string;
begin
    Assign(f, 'output1.txt');
    Rewrite(f);
    var n := ReadInteger('Введите n:');
    writeln(f,n); // запишем количество элементов
    for var i:=1 to n do writeln(f, (random(201)-100)/10);
    close (f);

    Reset(f);
    var m:integer;
    readln(f,m); // прочитаем количество элементов
    for var i:=1 to m do
        begin
            readln(f,a);
            writeln(StrToFloat(a):10:5);
        end;
    close(f);
end.
```

Посмотрим на созданный нами файл в блокноте:

```
output1 - Блокнот
Файл  Правка  Формат  Вид  Справка
10
-2.2
-6.5
-6.5
9.8
0.1
-7.4
9.4
-7
6.3
-9.9
```

Количество элементов